

Push Inv to QB

#Prep Invoice

Set Field [Main::gResult; PCQB_RqNew("InvoiceAdd")]

Set Field [Main::gResult; PCQB_RqAddFieldWithValue("CustomerRef::ListID" ; Main::ListID)]

Set Field [Main::gResult; PCQB_RqAddFieldWithValue("TxnDate"; Main::TxnDate) &
PCQB_RqAddFieldWithValue("PONumber"; Main::PONumber) &
PCQB_RqAddFieldWithValue("DueDate"; Main::DueDate) &
PCQB_RqAddFieldWithValue("ShipDate"; Main::ShipDate)]

#Prep Invoice Line Items

Go to Portal Row

[Select; First]

Loop

Set Field [Main::gResult; PCQB_RqAddRelatedRecord("InvoiceLineAdd" ; "")]

Set Field [Main::gResult; PCQB_RqAddFieldWithValue("ItemRef::FullName" ; LI::ItemNo) &
PCQB_RqAddFieldWithValue("Desc" ; LI::Desc) &
PCQB_RqAddFieldWithValue("Quantity" ; LI::Quantity) &
PCQB_RqAddFieldWithValue("Rate" ; LI::Rate) &
PCQB_RqAddFieldWithValue("ClassRef::FullName"; LI::ClassRefFullName) &
PCQB_RqAddFieldWithValue("ServiceDate"; LI::ServiceDate)]

Set Field [Main::gResult; PCQB_RqCloseRelatedRecord]

Go to Portal Row

[Select; Next; Exit after last]

Exit Loop If [not IsValid(LI::Quantity)]

End Loop

#Create Invoice (Begin Session, Execute and Close Session)

Set Field [Main::gResult; PCQB_BeginSession ("";"")]

Set Field [Main::gResult; PCQB_RqExecute]

Set Field [Main::gResult; PCQB_EndSession]

#Capture the QB Invoice Number

Set Field [Main::gResult; PCQB_RsOpenFirstRecord]

Set Field [Main::InvoiceNumber; PCQB_RsGetFirstFieldValue("RefNumber")]